
1. Overview

This document lists all the essencial changes needed to correct problems in the TTCN implementation of test case 22.5.18 which is part of the NBIOT test suite in the ‘iwd-TTCN3-B2016-06_D17wk23’ ATS delivery. 

With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence. 

Contact:
Yongsheng MA

mayongsheng@starpointcomm.com

Tel. +8615117939410

1.1 Verification Test Summary
Test Case:
TC_22_5_18
ATS Version:
iwd-TTCN3-B2016-06_D17wk23
System Simulator used:
Starpoint SP8630

UE used:
Hisilicon Hi2110
Verification Status:
PASS
2. Corrections required

2.1 Change 1 – Correction to function ‘f_TC_22_5_18_NBIOT’
	Function name
	f_TC_22_5_18_NBIOT ()

	Reason for change
	1. Since Ncell2 has the same TA with Ncell1, the UE will not transimit TAU request at step 24. The related prose CR will be present in the next RAN#76 meeting. 
2. Since UE has a valid context, it will send GUTI and TAI when transimits attach request and TAU request message at step1, step 36, step 70 and step 102. However, the function these steps use does not include GUTI or TAI.
3. According to cl.8.2.29.18 in TS24.301, if the UE requests "SMS only" or CIoT EPS optimizations,UE shall include additional update type IE in TAU request message at step 24, step 93.

4. This test case requires UE configured PSM, therefore, UE may request T3324 at step 1, step 24 and step 36.

5. After step 55B ad step 56, UE reselect from Ncell 1 to Ncell 11 without release. The NWK need to reset the RBs on Ncell 1 before step 70 when UE performs attach on Ncell1.
6. T3324 should be started after the value has been accepted and assinged by NWK.
7. The TAU accept message sent in step 58 and step 94 should include the PLMN that the serving cell uses.

8. Step 116 can’t be achieved since the UE is in RRC connected state after step 115. The related prose CR will be present in the next RAN#76 meeting.

	Summary of change
	1. Use Ncell11 instead of Ncell 2 in this test case.
2. Add corrected TAI value for this test case. And modiy the function that step1, step 36, step 70 and step 102 use. Please see more details in change 2.
3. Add additional update type IE in TAU request message at step 24, step 93.

4. Making T3324 value allowed in TAU request message at step 1, step 24 and step 36.

5. Reset RBs of Ncell1 before step 68.
6. T3324 should be started after step 96.

7. Modify the PLMN to PLMN1 that the serving cell uses for the TAU accept message at step 58 and step 94.

8. Add RRC connection release after step 115.

	TTCN module
	NBIOT_NAS_Attach.ttcn

	MCC160 Comment
	


Before change

   function fl_TC_22_5_18_Body() runs on NBIOT_PTC

  {

    // ==========================================================================

    // initialize constant test case data structures

    // ==========================================================================

    var NAS_MSG_Indication_Type v_NasInd;

    var NBIOT_PDN_Type v_PDN;

    var template (omit) EPS_BearerContextStatus v_EPS_ContextStatus;

    var GutiParameters_Type v_Guti_Params1 := f_NBIOT_CellInfo_GetGuti (nbiot_Cell1);

    var GutiParameters_Type v_Guti_Params2 := f_NBIOT_CellInfo_GetGuti (nbiot_Cell2);

    var GutiParameters_Type v_Guti1:= f_NBIOT_CellInfo_GetGuti(nbiot_Cell1);

    var GutiParameters_Type v_Guti_Params3 := f_CreateGutiParamsWithUpdatedSTmsi(v_Guti_Params1);

    var EPS_BearerIdentity v_EpsDefaultBearerId := tsc_EpsDefaultBearerId;

    var template (value) NB_CellPowerList_Type v_CellPowerList;

    var template (value) GPRS_Timer2 v_T3324Value2Mins := cs_GPRS_Timer2_3_IEI('6A'O, tsc_GprsTimerUnit_1min, '00010'B);  //value = 2 mins

    var float v_NoReactionTime := 30.0;

    timer t_T3324 := f_NBIOT_SetTimerToleranceMax(nasTimer, 120.0);

    //@siclog "Steps 1-14b1" siclog@

    //Steps 1 - 14b1 of the generic procedure specified in TS 36.508 subclause 8.1.5.2.3 are performed

    f_UT_SwitchOnUE(UT, true); //Step1 - Switch on the UE

    fl_NBIOT_InitialRegistration2to13_eDRX_PSM(nbiot_Cell1, omit, cs_EDRXParams('0000'B,'0011'B)); //Table 22.5.18.3.3-3: Steps 14bx handled on next step of prose

      v_PDN := f_NBIOT_MobileInfo_GetAssignedPdn (0);  // This is registration, so this is the first PDN

    //@siclog "Step 15" siclog@

    //Check: Does the UE transmit an ATTACH COMPLETE message ?

    f_NBIOT_NAS_AttachComplete_CP (nbiot_Cell1, v_EpsDefaultBearerId, v_PDN.assignedPdnType);

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 15");

    //@siclog "Step 16" siclog@

    //The SS releases the RRC connection.

    f_NBIOT_RRC_ConnectionRelease(nbiot_Cell1);

    //@siclog "Step 17" siclog@

    // SS transmits a Paging message to the UE in a valid paging occasion in paging hyperframe as per Idle eDRX

    f_NBIOT_UE_Page_eDRX (nbiot_Cell1,

                          cs_NB_Paging_OneRecord(cs_PagingUE_Identity_S_TMSI(v_Guti_Params1.MME_Code, v_Guti_Params1.M_TMSI)),

                          0,  // PTW index value (2.56 sec)

                          3); // eDRX index value (40.96 sec)

    //@siclog "Step 18 - 21B" siclog@

    // Steps 18-19: Steps 2 to 3 of the NB-IoT Control Plane CIoT MT user data transfer non-SMS transport procedure (TS 36.508 8.1.5A.2.3) are executed.

    // Steps 20-21: Void

    // Step 21A : Check: Does the UE transmit a CONTROL PLANE SERVICE REQUEST according to step 4a1a1 or 4a1b1 in the generic procedure in TS 36.508 8.1.5A.2.3?

    // Step 21B: Steps 4a1b1 to 5 or only step 5 of the NB-IoT Control Plane CIoT MT user data transfer non-SMS transport procedure (TS 36.508 8.1.5A.2.3) are executed

    f_NBIOT_508Check_CP_ResponseToPagingForMTAccess_Step2_5(nbiot_Cell1);

    //@siclog "Step 22" siclog@

    //The SS releases the RRC connection.

    f_NBIOT_RRC_ConnectionRelease(nbiot_Cell1);

    //@siclog "Step 23" siclog@

    //The SS configures:

    //- Ncell 1 as a "Non-Suitable cell".

    //- Ncell 2 as the "Serving cell".

    v_CellPowerList := {

      cs_NB_CellPower(nbiot_Cell1, tsc_NonSuitableNBIOTCellRS_EPRE),

      cs_NB_CellPower(nbiot_Cell2, tsc_ServingCellRS_EPRE)

    };

    f_NBIOT_SetCellPowerList(v_CellPowerList);

    //@siclog "Step 24" siclog@

    // The UE transmits a TRACKING AREA UPDATE REQUEST message including an Extended DRX parameters IE

    v_NasInd := f_NBIOT_RRC_ConnEst_DefWithNas(nbiot_Cell2,

                                               cr_EstablishmentCause_NB_Any,

                                               cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                                 cdr_TAU_Request_ExtdDrx_T3324 (tsc_EpsUpdate_TaUpdate,  // Table 22.5.18.3.3-5

                                                                                                f_NBIOT_SecurityKSIasme_Get(),

                                                                                                -,

                                                                                                omit,

                                                                                                cr_MobileIdAny,

                                                                                                cr_TAI_Any,

                                                                                                omit,

                                                                                                cr_ExtdDRXParamsAny)));

    v_EPS_ContextStatus := v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus; // Send this back in the Accept

    //@siclog "Step 25" siclog@

    // The SS transmits a TRACKING AREA UPDATE ACCEPT message without an Extended DRX parameters IE. GUTI-1 is included

    //@siclog "Step 26" siclog@

    //The UE transmits a TRACKING AREA UPDATE COMPLETE message

    f_NBIOT_TrackingAreaUpdate_Accept_Complete(nbiot_Cell2,

                                               CONTROL_PLANE,

                                               v_Guti1, // Table 22.5.18.3.3-6

                                               v_EPS_ContextStatus,

                                               v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType);

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 26");

    //@siclog "Step 27" siclog@

    //The SS releases the RRC connection.

    f_NBIOT_RRC_ConnectionRelease(nbiot_Cell2);

    //@siclog "Step 28 - 32" siclog@

    // Step 28: SS transmits a Paging-NB message to the UE in a valid paging occasion as per normal DRX

    // Steps 29- 30: Steps 2 to 3 of the NB-IoT Control Plane CIoT MT user data transfer non-SMS transport procedure (TS 36.508 8.1.5A.2.3) are executed.

    // Steps 31-32: Void

    // Step 32A: Check: Does the UE transmit a CONTROL PLANE SERVICE REQUEST according to step 4a1a1 or 4a1b1 in the generic procedure in TS 36.508 8.1.5A.2.3?

    // Step 32B: Steps 4a1b1 to 5 or only step 5 of the NB-IoT Control Plane CIoT MT user data transfer non-SMS transport procedure (TS 36.508 8.1.5A.2.3) are executed

    f_NBIOT_508Check_CP_ResponseToPagingForMTAccess(nbiot_Cell2);

    //@siclog "Step 33" siclog@

    // The UE is switched off - Note 1: This implies detaching of the UE, releasing of the RRC connection and resetting of the radio bearers at the SS side.

    f_NBIOT_UE_Detach_SwitchOffUe(nbiot_Cell1, CONTROL_PLANE, RRC_CONNECTED);

    //@siclog "Step 34" siclog@

    // Set the cell type of Ncell 1 to the "Serving cell"

    // Set the cell type of Ncell 2 to the "Non-Suitable cell".

    v_CellPowerList := {

      cs_NB_CellPower(nbiot_Cell1, tsc_ServingCellRS_EPRE),

      cs_NB_CellPower(nbiot_Cell2, tsc_NonSuitableNBIOTCellRS_EPRE)

    };

    f_NBIOT_SetCellPowerList(v_CellPowerList);

    //@siclog "Step 35" siclog@

    //The UE is switched on

    f_UT_SwitchOnUE(UT, true);

    //@siclog "Step 36-48b1" siclog@

    //Steps 2 - 14b1 of the generic procedure precified in TS 36.508 subclause 8.1.5.2.3 are performed

    fl_NBIOT_InitialRegistration2to13_eDRX_PSM(nbiot_Cell1); //Table 22.5.18.3.3-9: Step 14bx handled on next step of prose

    //@siclog "Step 49" siclog@

    // Check: Does the UE transmit an ATTACH COMPLETE message without an Extended DRX parameters IE?

    //Value of v_ESMMessage checked already from Step15

    f_NBIOT_NAS_AttachComplete_CP (nbiot_Cell1, v_EpsDefaultBearerId, v_PDN.assignedPdnType);

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 49");

    //@siclog "Step 50" siclog@

    //The SS releases the RRC connection.

    f_NBIOT_RRC_ConnectionRelease(nbiot_Cell1);

    //@siclog "Step 51 - 55B" siclog@

    // Step 51: SS transmits a Paging-NB message to the UE in a valid paging occasion as per normal DRX

    // Steps 52- 53: Steps 2 to 3 of the NB-IoT Control Plane CIoT MT user data transfer non-SMS transport procedure (TS 36.508 8.1.5A.2.3) are executed.

    // Steps 54-55: Void

    // Step 55A: Check: Does the UE transmit a CONTROL PLANE SERVICE REQUEST according to step 4a1a1 or 4a1b1 in the generic procedure in TS 36.508 8.1.5A.2.3?

    // Step 55B: Steps 4a1b1 to 5 or only step 5 of the NB-IoT Control Plane CIoT MT user data transfer non-SMS transport procedure (TS 36.508 8.1.5A.2.3) are executed

    f_NBIOT_508Check_CP_ResponseToPagingForMTAccess(nbiot_Cell1);

    //@siclog "Step 56" siclog@

    // Set the cell type of Ncell 1 to the "Non-Suitable cell".

    // Set the cell type of Ncell 2 to the "Serving cell"

    v_CellPowerList := {

      cs_NB_CellPower(nbiot_Cell1, tsc_NonSuitableNBIOTCellRS_EPRE),

      cs_NB_CellPower(nbiot_Cell2, tsc_ServingCellRS_EPRE)

    };

    f_NBIOT_SetCellPowerList(v_CellPowerList);

    //@siclog "Step 57" siclog@

    // The UE transmits a TRACKING AREA UPDATE REQUEST message including an Extended DRX parameters IE

    v_NasInd := f_NBIOT_TrackingAreaUpdate_Step1_3(nbiot_Cell2, ?);

    v_EPS_ContextStatus := v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus; // Send this back in the Accept

    //@siclog "Step 58" siclog@

    // The SS transmits a TRACKING AREA UPDATE ACCEPT message without an Extended DRX parameters IE. GUTI-2 is included

    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell2,

                                   tsc_SRB1bis,

                                   cs_TimingInfo_Now,

                                   cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                  cs_TAU_Accept_Common(tsc_EpsUpdate_TaUpdate,

                                                                              f_GutiParameters2MobileIdentity(tsc_IEI_Guti, v_Guti_Params2), //Table 22.5.18.3.3-12

                                                                              cds_TAIListNonConsecutive_tlv (tsc_PLMN2, {tsc_TAC2}),

                                                                              v_EPS_ContextStatus,

                                                                              omit,

                                                                              omit,

                                                                              f_GetNBIOTAdditionalUpdateResult(v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType, CONTROL_PLANE),

                                                                               -, -, -,

                                                                              f_GetNBIOTNetworkFeatureSupport(CONTROL_PLANE),  // @sic R5-170563 sic@

                                                                               -, -, -,

                                                                              omit,

                                                                              cs_EDRXParams('0000'B, '0011'B)))));

    //@siclog "Step 59" siclog@

    //The UE transmits a TRACKING AREA UPDATE COMPLETE message.

    SRB.receive(car_NB_SRB_NasPdu_IND(nbiot_Cell2,

                                      tsc_SRB1bis,

                                      cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                        cr_508_TAU_Complete)));

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 59");

    //@siclog "Step 60" siclog@

    //The SS releases the RRC connection.

    f_NBIOT_RRC_ConnectionRelease(nbiot_Cell2);

    //@siclog "Step 61" siclog@

    // SS transmits a Paging-NB message to the UE in a valid paging occasion in paging hyperframe as per Idle eDRX

    f_NBIOT_UE_Page_eDRX (nbiot_Cell2,

                          cs_NB_Paging_OneRecord(cs_PagingUE_Identity_S_TMSI(v_Guti_Params2.MME_Code, v_Guti_Params2.M_TMSI)),

                          0,  // PTW index value (2.56 sec)

                          3); // eDRX index value (40.96 sec)

    //@siclog "Step 62 - 65B" siclog@

    // Steps 62-63: Steps 2 to 3 of the NB-IoT Control Plane CIoT MT user data transfer non-SMS transport procedure (TS 36.508 8.1.5A.2.3) are executed.

    // Steps 64-65: Void

    // Step 65A : Check: Does the UE transmit a CONTROL PLANE SERVICE REQUEST according to step 4a1a1 or 4a1b1 in the generic procedure in TS 36.508 8.1.5A.2.3?

    // Step 65B: Steps 4a1b1 to 5 or only step 5 of the NB-IoT Control Plane CIoT MT user data transfer non-SMS transport procedure (TS 36.508 8.1.5A.2.3) are executed

    f_NBIOT_508Check_CP_ResponseToPagingForMTAccess_Step2_5(nbiot_Cell2);

    //@siclog "Step 66" siclog@

    //The SS releases the RRC connection.

    f_NBIOT_RRC_ConnectionRelease(nbiot_Cell2);

    //@siclog "Step 67" siclog@

    // The UE is switched off - Note 1: This implies detaching of the UE, releasing of the RRC connection and resetting of the radio bearers at the SS side.

    f_NBIOT_UE_Detach_SwitchOffUe(nbiot_Cell1, CONTROL_PLANE, RRC_IDLE);

    //@siclog "Step 68" siclog@

    // Set the cell type of Ncell 1 to the "Serving cell"

    // Set the cell type of Ncell 2 to the "Non-Suitable cell".

    v_CellPowerList := {

      cs_NB_CellPower(nbiot_Cell1, tsc_ServingCellRS_EPRE),

      cs_NB_CellPower(nbiot_Cell2, tsc_NonSuitableNBIOTCellRS_EPRE)

    };

    f_NBIOT_SetCellPowerList(v_CellPowerList);

    //@siclog "Step 69 - Void" siclog@

    //@siclog "Step 70-83b1" siclog@

    // Steps 1 - 14b1 of the generic procedure precified in TS 36.508 subclause 8.1.5.2.3 is performed: Step 14bX handled on next step of prose

    f_UT_SwitchOnUE(UT, true); //Step1 - Switch on the UE

    fl_NBIOT_InitialRegistration2to13_eDRX_PSM(nbiot_Cell1, v_T3324Value2Mins, cs_EDRXParams('0000'B,'0011'B)); //Table 22.5.18.3.3-14

    //@siclog "Step 84" siclog@

    //Check: Does the UE transmit an ATTACH COMPLETE?

    f_NBIOT_NAS_AttachComplete_CP (nbiot_Cell1, v_EpsDefaultBearerId, v_PDN.assignedPdnType);

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 84");

    //@siclog "Step 85" siclog@

    //The SS releases the RRC connection.

    f_NBIOT_RRC_ConnectionRelease(nbiot_Cell1);

    //@siclog "Step 86" siclog@

    // SS transmits a Paging-NB message to the UE in a valid paging occasion in paging hyperframe as per Idle eDRX

    f_NBIOT_UE_Page_eDRX (nbiot_Cell1,

                          cs_NB_Paging_OneRecord(cs_PagingUE_Identity_S_TMSI(v_Guti_Params1.MME_Code, v_Guti_Params1.M_TMSI)),

                          0,  // PTW index value (2.56 sec)

                          3); // eDRX index value (40.96 sec)

    //@siclog "Step 87 - 90B" siclog@

    // Steps 87-88: Steps 2 to 3 of the NB-IoT Control Plane CIoT MT user data transfer non-SMS transport procedure (TS 36.508 8.1.5A.2.3) are executed.

    // Steps 89-90: Void

    // Step 90A : Check: Does the UE transmit a CONTROL PLANE SERVICE REQUEST according to step 4a1a1 or 4a1b1 in the generic procedure in TS 36.508 8.1.5A.2.3?

    // Step 90B: Steps 4a1b1 to 5 or only step 5 of the NB-IoT Control Plane CIoT MT user data transfer non-SMS transport procedure (TS 36.508 8.1.5A.2.3) are executed

    f_NBIOT_508Check_CP_ResponseToPagingForMTAccess_Step2_5(nbiot_Cell1);

    //@siclog "Step 91" siclog@

    //The SS releases the RRC connection.

    f_NBIOT_RRC_ConnectionRelease(nbiot_Cell1);

    //@siclog "Step 92" siclog@

    // Set the cell type of Ncell 1 to the "Non-Suitable cell"

    // Set the cell type of Ncell 2 to the "Serving cell"

    v_CellPowerList := {

      cs_NB_CellPower(nbiot_Cell2, tsc_ServingCellRS_EPRE),

      cs_NB_CellPower(nbiot_Cell1, tsc_NonSuitableNBIOTCellRS_EPRE)

    };

    f_NBIOT_SetCellPowerList(v_CellPowerList);

    //@siclog "Step 93" siclog@

    // The UE transmits a TRACKING AREA UPDATE REQUEST message including an Extended DRX parameters IE and the T3324 IE set to two minutes

    v_NasInd := f_NBIOT_RRC_ConnEst_DefWithNas(nbiot_Cell2,

                                               cr_EstablishmentCause_NB_Any,

                                               cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                                 cdr_TAU_Request_ExtdDrx_T3324 (tsc_EpsUpdate_TaUpdate,

                                                                                                f_NBIOT_SecurityKSIasme_Get(),

                                                                                                -,

                                                                                                omit,

                                                                                                cr_MobileIdAny,

                                                                                                cr_TAI_Any,

                                                                                                v_T3324Value2Mins, // Table 22.5.18.3.3-16

                                                                                                cr_ExtdDRXParamsAny)));

    v_EPS_ContextStatus := v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus; // Send this back in the Accept

    t_T3324.start;
    //@siclog "Step 94" siclog@

    // The SS transmits a TRACKING AREA UPDATE ACCEPT message without an Extended DRX parameters IE but with the T3324 IE set to 2 minutes. GUTI-3 is included

    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell2,

                                   tsc_SRB1bis,

                                   cs_TimingInfo_Now,

                                   cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                  cs_TAU_Accept_Common(tsc_EpsUpdate_TaUpdate, // Table 22.5.18.3.3-17

                                                                              f_GutiParameters2MobileIdentity(tsc_IEI_Guti, v_Guti_Params3),

                                                                              cds_TAIListNonConsecutive_tlv (tsc_PLMN2, {tsc_TAC2}),

                                                                              v_EPS_ContextStatus,

                                                                              omit,//Lai

                                                                              omit,//MSId

                                                                              f_GetNBIOTAdditionalUpdateResult(v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType, CONTROL_PLANE),

                                                                              -, -, -,

                                                                              f_GetNBIOTNetworkFeatureSupport(CONTROL_PLANE),  // @sic R5-170563 sic@

                                                                              -, -, -,

                                                                              v_T3324Value2Mins)))); //T3324

    //@siclog "Step 95" siclog@

    //The UE transmits a TRACKING AREA UPDATE COMPLETE message

    SRB.receive(car_NB_SRB_NasPdu_IND(nbiot_Cell2,

                                      tsc_SRB1bis,

                                      cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                        cr_508_TAU_Complete)));

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 95");

    //@siclog "Step 96" siclog@

    //The SS releases the RRC connection.

    f_NBIOT_RRC_ConnectionRelease(nbiot_Cell2);

    t_T3324.timeout;

    //@siclog "Step 97" siclog@

    // When the T3324 timer expires the SS sends Paging message including a matched identity in a valid paging occasion as per normal DRX

    //@siclog "Step 98" siclog@

    // Check: Does the UE respond to the paging message?

    f_NBIOT_UE_Page_NoReaction (nbiot_Cell2,

                                cs_NB_Paging_OneRecord(cs_PagingUE_Identity_S_TMSI(v_Guti_Params3.MME_Code, v_Guti_Params3.M_TMSI)),

                                v_NoReactionTime);

    //@siclog "Step 99" siclog@

    // The UE is switched off

    f_UT_SwitchOffUE (UT, false);

    //@siclog "Step 100" siclog@

    // Set the cell type of Ncell 1 to the "Serving cell"

    // Set the cell type of Ncell 2 to the "Non-Suitable cell".

    v_CellPowerList := {

      cs_NB_CellPower(nbiot_Cell1, tsc_ServingCellRS_EPRE),

      cs_NB_CellPower(nbiot_Cell2, tsc_NonSuitableNBIOTCellRS_EPRE)

    };

    f_NBIOT_SetCellPowerList(v_CellPowerList);

    //@siclog "Step 101" siclog@

    f_UT_SwitchOnUE(UT, true); //Step1 - Switch on the UE

    //@siclog "Steps 102-114b1" siclog@

    // Steps 2 - 14b1 of the generic procedure precified in TS 36.508 subclause 8.1.5.2.3 is performed: Step 14bX handled on next step of prose

    fl_NBIOT_InitialRegistration2to13_eDRX_PSM(nbiot_Cell1, v_T3324Value2Mins, cs_EDRXParams('0000'B,'0011'B), true); //Table 22.5.18.3.3-19

    //@siclog "Step 115" siclog@

    // Check: Does the UE transmit an ATTACH COMPLETE message including an Extended DRX parameters IE but without the T3324 IE?

    f_NBIOT_NAS_AttachComplete_CP (nbiot_Cell1, v_EpsDefaultBearerId, v_PDN.assignedPdnType);

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 115");

    //@siclog "Step 116" siclog@

    // SS sends a Paging message to the UE in a valid paging occasion in paging hyperframe as per Idle eDRX

    f_NBIOT_UE_Page_eDRX (nbiot_Cell1,

                          cs_NB_Paging_OneRecord(cs_PagingUE_Identity_S_TMSI(v_Guti_Params1.MME_Code, v_Guti_Params1.M_TMSI)),

                          0,  // PTW index value (2.56 sec)

                          3); // eDRX index value (40.96 sec)

    //@siclog "Step 117 - 120B" siclog@

    // Steps 117-118: Steps 2 to 3 of the NB-IoT Control Plane CIoT MT user data transfer non-SMS transport procedure (TS 36.508 8.1.5A.2.3) are executed.

    // Steps 119-120: Void

    // Step 120A : Check: Does the UE transmit a CONTROL PLANE SERVICE REQUEST according to step 4a1a1 or 4a1b1 in the generic procedure in TS 36.508 8.1.5A.2.3?

    // Step 120B: Steps 4a1b1 to 5 or only step 5 of the NB-IoT Control Plane CIoT MT user data transfer non-SMS transport procedure (TS 36.508 8.1.5A.2.3) are executed

    f_NBIOT_508Check_CP_ResponseToPagingForMTAccess_Step2_5(nbiot_Cell1);

    //@siclog "Step 121" siclog@

    //The SS releases the RRC connection.

    f_NBIOT_RRC_ConnectionRelease(nbiot_Cell1);

  } // function fl_tc_22_5_18_Body
After change

    function fl_TC_22_5_18_Body() runs on NBIOT_PTC

  {

    // ==========================================================================

    // initialize constant test case data structures

    // ==========================================================================

    var NAS_MSG_Indication_Type v_NasInd;

    var NBIOT_PDN_Type v_PDN;

    var template (omit) EPS_BearerContextStatus v_EPS_ContextStatus;

    var GutiParameters_Type v_Guti_Params1 := f_NBIOT_CellInfo_GetGuti (nbiot_Cell1);

    var GutiParameters_Type v_Guti_Params2 := f_NBIOT_CellInfo_GetGuti (nbiot_Cell11);

    var GutiParameters_Type v_Guti1:= f_NBIOT_CellInfo_GetGuti(nbiot_Cell1);

    var GutiParameters_Type v_Guti_Params3 := f_CreateGutiParamsWithUpdatedSTmsi(v_Guti_Params1);

    var EPS_BearerIdentity v_EpsDefaultBearerId := tsc_EpsDefaultBearerId;

    var template (value) NB_CellPowerList_Type v_CellPowerList;

    var template (value) GPRS_Timer2 v_T3324Value2Mins := cs_GPRS_Timer2_3_IEI('6A'O, tsc_GprsTimerUnit_1min, '00010'B);  //value = 2 mins

    var float v_NoReactionTime := 30.0;

    timer t_T3324 := f_NBIOT_SetTimerToleranceMax(nasTimer, 120.0);

    var template TrackingAreaId v_Tai1 := cr_TAI(f_Asn2Nas_PlmnId(v_Guti_Params1.PLMN_Identity), bit2oct(f_NBIOT_CellInfo_GetTAC(nbiot_Cell1)));
    var template TrackingAreaId v_Tai2 := cr_TAI(f_Asn2Nas_PlmnId(v_Guti_Params2.PLMN_Identity), bit2oct(f_NBIOT_CellInfo_GetTAC(nbiot_Cell11)));
    var template TrackingAreaId v_Tai3 := cr_TAI(f_Asn2Nas_PlmnId(v_Guti_Params3.PLMN_Identity), bit2oct(f_NBIOT_CellInfo_GetTAC(nbiot_Cell11)));
    //@siclog "Steps 1-14b1" siclog@

    //Steps 1 - 14b1 of the generic procedure specified in TS 36.508 subclause 8.1.5.2.3 are performed

    f_UT_SwitchOnUE(UT, true); //Step1 - Switch on the UE

    fl_NBIOT_InitialRegistration2to13_eDRX_PSM(nbiot_Cell1, *, cs_EDRXParams('0000'B,'0011'B), -, v_Guti_Params1, v_Tai1); //Table 22.5.18.3.3-14

    v_PDN := f_NBIOT_MobileInfo_GetAssignedPdn (0);  // This is registration, so this is the first PDN

    //@siclog "Step 15" siclog@

    //Check: Does the UE transmit an ATTACH COMPLETE message ?

    f_NBIOT_NAS_AttachComplete_CP (nbiot_Cell1, v_EpsDefaultBearerId, v_PDN.assignedPdnType);

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 15");

    //@siclog "Step 16" siclog@

    //The SS releases the RRC connection.

    f_NBIOT_RRC_ConnectionRelease(nbiot_Cell1);

    //@siclog "Step 17" siclog@

    // SS transmits a Paging message to the UE in a valid paging occasion in paging hyperframe as per Idle eDRX

    f_NBIOT_UE_Page_eDRX (nbiot_Cell1,

                          cs_NB_Paging_OneRecord(cs_PagingUE_Identity_S_TMSI(v_Guti_Params1.MME_Code, v_Guti_Params1.M_TMSI)),

                          0,  // PTW index value (2.56 sec)

                          3); // eDRX index value (40.96 sec)

    //@siclog "Step 18 - 21B" siclog@

    // Steps 18-19: Steps 2 to 3 of the NB-IoT Control Plane CIoT MT user data transfer non-SMS transport procedure (TS 36.508 8.1.5A.2.3) are executed.

    // Steps 20-21: Void

    // Step 21A : Check: Does the UE transmit a CONTROL PLANE SERVICE REQUEST according to step 4a1a1 or 4a1b1 in the generic procedure in TS 36.508 8.1.5A.2.3?

    // Step 21B: Steps 4a1b1 to 5 or only step 5 of the NB-IoT Control Plane CIoT MT user data transfer non-SMS transport procedure (TS 36.508 8.1.5A.2.3) are executed

    f_NBIOT_508Check_CP_ResponseToPagingForMTAccess_Step2_5(nbiot_Cell1);

    //@siclog "Step 22" siclog@

    //The SS releases the RRC connection.

    f_NBIOT_RRC_ConnectionRelease(nbiot_Cell1);

    //@siclog "Step 23" siclog@

    //The SS configures:

    //- Ncell 1 as a "Non-Suitable cell".

    //- Ncell 11 as the "Serving cell".

    v_CellPowerList := {

      cs_NB_CellPower(nbiot_Cell1, tsc_NonSuitableNBIOTCellRS_EPRE),

      cs_NB_CellPower(nbiot_Cell11, tsc_ServingCellRS_EPRE)

    };

    f_NBIOT_SetCellPowerList(v_CellPowerList);

    //@siclog "Step 24" siclog@

    // The UE transmits a TRACKING AREA UPDATE REQUEST message including an Extended DRX parameters IE

    v_NasInd := f_NBIOT_RRC_ConnEst_DefWithNas(nbiot_Cell11,

                                               cr_EstablishmentCause_NB_Any,

                                               cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                                 cdr_TAU_Request_ExtdDrx_T3324 (tsc_EpsUpdate_TaUpdate,  // Table 22.5.18.3.3-5

                                                                                                f_NBIOT_SecurityKSIasme_Get(),

                                                                                                -,

                                                                                                cr_AdditionalUpdateType_CIOT,//omit,

                                                                                                cr_MobileIdAny,

                                                                                                cr_TAI_Any,

                                                                                                *,

                                                                                                cr_ExtdDRXParamsAny)));

    v_EPS_ContextStatus := v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus; // Send this back in the Accept

    //@siclog "Step 25" siclog@

    // The SS transmits a TRACKING AREA UPDATE ACCEPT message without an Extended DRX parameters IE. GUTI-1 is included

    //@siclog "Step 26" siclog@

    //The UE transmits a TRACKING AREA UPDATE COMPLETE message

    f_NBIOT_TrackingAreaUpdate_Accept_Complete(nbiot_Cell11,

                                               CONTROL_PLANE,

                                               v_Guti1, // Table 22.5.18.3.3-6

                                               v_EPS_ContextStatus,

                                               v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType);

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 26");

    //@siclog "Step 27" siclog@

    //The SS releases the RRC connection.

    f_NBIOT_RRC_ConnectionRelease(nbiot_Cell11);

    //@siclog "Step 28 - 32" siclog@

    // Step 28: SS transmits a Paging-NB message to the UE in a valid paging occasion as per normal DRX

    // Steps 29- 30: Steps 2 to 3 of the NB-IoT Control Plane CIoT MT user data transfer non-SMS transport procedure (TS 36.508 8.1.5A.2.3) are executed.

    // Steps 31-32: Void

    // Step 32A: Check: Does the UE transmit a CONTROL PLANE SERVICE REQUEST according to step 4a1a1 or 4a1b1 in the generic procedure in TS 36.508 8.1.5A.2.3?

    // Step 32B: Steps 4a1b1 to 5 or only step 5 of the NB-IoT Control Plane CIoT MT user data transfer non-SMS transport procedure (TS 36.508 8.1.5A.2.3) are executed

    f_NBIOT_508Check_CP_ResponseToPagingForMTAccess(nbiot_Cell11);

    //@siclog "Step 33" siclog@

    // The UE is switched off

    f_NBIOT_UE_Detach_SwitchOffUe(nbiot_Cell11, CONTROL_PLANE, RRC_CONNECTED); //FFS: prose update triggered//cell1

    //@siclog "Step 34" siclog@

    // Set the cell type of Ncell 1 to the "Serving cell"

    // Set the cell type of Ncell 2 to the "Non-Suitable cell".

    v_CellPowerList := {

      cs_NB_CellPower(nbiot_Cell1, tsc_ServingCellRS_EPRE),

      cs_NB_CellPower(nbiot_Cell11, tsc_NonSuitableNBIOTCellRS_EPRE)

    };

    f_NBIOT_SetCellPowerList(v_CellPowerList);

    //@siclog "Step 35" siclog@

    //The UE is switched on

    f_UT_SwitchOnUE(UT, true);

    //@siclog "Step 36-48b1" siclog@

    //Steps 2 - 14b1 of the generic procedure precified in TS 36.508 subclause 8.1.5.2.3 are performed

    fl_NBIOT_InitialRegistration2to13_eDRX_PSM(nbiot_Cell1, *, omit, -, v_Guti_Params2, v_Tai2); //Table 22.5.18.3.3-14

    //@siclog "Step 49" siclog@

    // Check: Does the UE transmit an ATTACH COMPLETE message without an Extended DRX parameters IE?

    //Value of v_ESMMessage checked already from Step15

    f_NBIOT_NAS_AttachComplete_CP (nbiot_Cell1, v_EpsDefaultBearerId, v_PDN.assignedPdnType);

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 49");

    //@siclog "Step 50" siclog@

    //The SS releases the RRC connection.

    f_NBIOT_RRC_ConnectionRelease(nbiot_Cell1);

    //@siclog "Step 51 - 55B" siclog@

    // Step 51: SS transmits a Paging-NB message to the UE in a valid paging occasion as per normal DRX

    // Steps 52- 53: Steps 2 to 3 of the NB-IoT Control Plane CIoT MT user data transfer non-SMS transport procedure (TS 36.508 8.1.5A.2.3) are executed.

    // Steps 54-55: Void

    // Step 55A: Check: Does the UE transmit a CONTROL PLANE SERVICE REQUEST according to step 4a1a1 or 4a1b1 in the generic procedure in TS 36.508 8.1.5A.2.3?

    // Step 55B: Steps 4a1b1 to 5 or only step 5 of the NB-IoT Control Plane CIoT MT user data transfer non-SMS transport procedure (TS 36.508 8.1.5A.2.3) are executed

    f_NBIOT_508Check_CP_ResponseToPagingForMTAccess(nbiot_Cell1);

    //@siclog "Step 56" siclog@

    // Set the cell type of Ncell 1 to the "Non-Suitable cell".

    // Set the cell type of Ncell 11 to the "Serving cell"

    v_CellPowerList := {

      cs_NB_CellPower(nbiot_Cell1, tsc_NonSuitableNBIOTCellRS_EPRE),

      cs_NB_CellPower(nbiot_Cell11, tsc_ServingCellRS_EPRE)

    };

    f_NBIOT_SetCellPowerList(v_CellPowerList);

    //@siclog "Step 57" siclog@

    // The UE transmits a TRACKING AREA UPDATE REQUEST message including an Extended DRX parameters IE

    v_NasInd := f_NBIOT_TrackingAreaUpdate_Step1_3(nbiot_Cell11, ?);

    v_EPS_ContextStatus := v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus; // Send this back in the Accept

    //@siclog "Step 58" siclog@

    // The SS transmits a TRACKING AREA UPDATE ACCEPT message without an Extended DRX parameters IE. GUTI-2 is included

    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell11,

                                   tsc_SRB1bis,

                                   cs_TimingInfo_Now,

                                   cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                  cs_TAU_Accept_Common(tsc_EpsUpdate_TaUpdate,

                                                                              f_GutiParameters2MobileIdentity(tsc_IEI_Guti, v_Guti_Params2), //Table 22.5.18.3.3-12

                                                                              cds_TAIListNonConsecutive_tlv (tsc_PLMN1, {tsc_TAC2}),//tsc_PLMN2 TTCN CR

                                                                              v_EPS_ContextStatus,

                                                                              omit,

                                                                              omit,

                                                                              f_GetNBIOTAdditionalUpdateResult(v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType, CONTROL_PLANE),

                                                                               -, -, -,

                                                                              f_GetNBIOTNetworkFeatureSupport(CONTROL_PLANE),  // @sic R5-170563 sic@

                                                                               -, -, -,

                                                                              omit,

                                                                              cs_EDRXParams('0000'B, '0011'B)))));

    //@siclog "Step 59" siclog@

    //The UE transmits a TRACKING AREA UPDATE COMPLETE message.

    SRB.receive(car_NB_SRB_NasPdu_IND(nbiot_Cell11,

                                      tsc_SRB1bis,

                                      cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                        cr_508_TAU_Complete)));

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 59");

    //@siclog "Step 60" siclog@

    //The SS releases the RRC connection.

    f_NBIOT_RRC_ConnectionRelease(nbiot_Cell11);

    //@siclog "Step 61" siclog@

    // SS transmits a Paging-NB message to the UE in a valid paging occasion in paging hyperframe as per Idle eDRX

    f_NBIOT_UE_Page_eDRX (nbiot_Cell11,

                          cs_NB_Paging_OneRecord(cs_PagingUE_Identity_S_TMSI(v_Guti_Params2.MME_Code, v_Guti_Params2.M_TMSI)),

                          0,  // PTW index value (2.56 sec)

                          3); // eDRX index value (40.96 sec)

    //@siclog "Step 62 - 65B" siclog@

    // Steps 62-63: Steps 2 to 3 of the NB-IoT Control Plane CIoT MT user data transfer non-SMS transport procedure (TS 36.508 8.1.5A.2.3) are executed.

    // Steps 64-65: Void

    // Step 65A : Check: Does the UE transmit a CONTROL PLANE SERVICE REQUEST according to step 4a1a1 or 4a1b1 in the generic procedure in TS 36.508 8.1.5A.2.3?

    // Step 65B: Steps 4a1b1 to 5 or only step 5 of the NB-IoT Control Plane CIoT MT user data transfer non-SMS transport procedure (TS 36.508 8.1.5A.2.3) are executed

    f_NBIOT_508Check_CP_ResponseToPagingForMTAccess_Step2_5(nbiot_Cell11);

    //@siclog "Step 66" siclog@

    //The SS releases the RRC connection.

    f_NBIOT_RRC_ConnectionRelease(nbiot_Cell11);

    //@siclog "Step 67" siclog@

    // The UE is switched off

    f_NBIOT_UE_Detach_SwitchOffUe(nbiot_Cell11, CONTROL_PLANE, RRC_IDLE); //FFS: prose update triggered

    //@siclog "Step 68" siclog@

    // Set the cell type of Ncell 1 to the "Serving cell"

    // Set the cell type of Ncell 11 to the "Non-Suitable cell".

    f_NBIOT_SS_SRBs_DRBs_Reset (nbiot_Cell1, CONTROL_PLANE);//TTCN CR!!

    v_CellPowerList := {

      cs_NB_CellPower(nbiot_Cell1, tsc_ServingCellRS_EPRE),

      cs_NB_CellPower(nbiot_Cell11, tsc_NonSuitableNBIOTCellRS_EPRE)

    };

    f_NBIOT_SetCellPowerList(v_CellPowerList);

    //@siclog "Step 69 - Void" siclog@

    //@siclog "Step 70-83b1" siclog@

    // Steps 1 - 14b1 of the generic procedure precified in TS 36.508 subclause 8.1.5.2.3 is performed: Step 14bX handled on next step of prose

    f_UT_SwitchOnUE(UT, true); //Step1 - Switch on the UE

    fl_NBIOT_InitialRegistration2to13_eDRX_PSM(nbiot_Cell1, v_T3324Value2Mins, cs_EDRXParams('0000'B,'0011'B), -, v_Guti_Params2, v_Tai2); //Table 22.5.18.3.3-14

    //@siclog "Step 84" siclog@

    //Check: Does the UE transmit an ATTACH COMPLETE?

    f_NBIOT_NAS_AttachComplete_CP (nbiot_Cell1, v_EpsDefaultBearerId, v_PDN.assignedPdnType);

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 84");

    //@siclog "Step 85" siclog@

    //The SS releases the RRC connection.

    f_NBIOT_RRC_ConnectionRelease(nbiot_Cell1);

    //@siclog "Step 86" siclog@

    // SS transmits a Paging-NB message to the UE in a valid paging occasion in paging hyperframe as per Idle eDRX

    f_NBIOT_UE_Page_eDRX (nbiot_Cell1,

                          cs_NB_Paging_OneRecord(cs_PagingUE_Identity_S_TMSI(v_Guti_Params1.MME_Code, v_Guti_Params1.M_TMSI)),

                          0,  // PTW index value (2.56 sec)

                          3); // eDRX index value (40.96 sec)

    //@siclog "Step 87 - 90B" siclog@

    // Steps 87-88: Steps 2 to 3 of the NB-IoT Control Plane CIoT MT user data transfer non-SMS transport procedure (TS 36.508 8.1.5A.2.3) are executed.

    // Steps 89-90: Void

    // Step 90A : Check: Does the UE transmit a CONTROL PLANE SERVICE REQUEST according to step 4a1a1 or 4a1b1 in the generic procedure in TS 36.508 8.1.5A.2.3?

    // Step 90B: Steps 4a1b1 to 5 or only step 5 of the NB-IoT Control Plane CIoT MT user data transfer non-SMS transport procedure (TS 36.508 8.1.5A.2.3) are executed

    f_NBIOT_508Check_CP_ResponseToPagingForMTAccess_Step2_5(nbiot_Cell1);

    //@siclog "Step 91" siclog@

    //The SS releases the RRC connection.

    f_NBIOT_RRC_ConnectionRelease(nbiot_Cell1);

    //@siclog "Step 92" siclog@

    // Set the cell type of Ncell 1 to the "Non-Suitable cell"

    // Set the cell type of Ncell 11 to the "Serving cell"

    v_CellPowerList := {

      cs_NB_CellPower(nbiot_Cell11, tsc_ServingCellRS_EPRE),

      cs_NB_CellPower(nbiot_Cell1, tsc_NonSuitableNBIOTCellRS_EPRE)

    };

    f_NBIOT_SetCellPowerList(v_CellPowerList);

    //@siclog "Step 93" siclog@

    // The UE transmits a TRACKING AREA UPDATE REQUEST message including an Extended DRX parameters IE and the T3324 IE set to two minutes

    v_NasInd := f_NBIOT_RRC_ConnEst_DefWithNas(nbiot_Cell11,

                                               cr_EstablishmentCause_NB_Any,

                                               cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                                 cdr_TAU_Request_ExtdDrx_T3324 (tsc_EpsUpdate_TaUpdate,

                                                                                                f_NBIOT_SecurityKSIasme_Get(),

                                                                                                -,

                                                                                                cr_AdditionalUpdateType_CIOT,

                                                                                                cr_MobileIdAny,

                                                                                                cr_TAI_Any,

                                                                                                v_T3324Value2Mins, // Table 22.5.18.3.3-16

                                                                                                cr_ExtdDRXParamsAny)));

    v_EPS_ContextStatus := v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus; // Send this back in the Accept

    //@siclog "Step 94" siclog@

    // The SS transmits a TRACKING AREA UPDATE ACCEPT message without an Extended DRX parameters IE but with the T3324 IE set to 2 minutes. GUTI-3 is included

    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell11,

                                   tsc_SRB1bis,

                                   cs_TimingInfo_Now,

                                   cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                  cs_TAU_Accept_Common(tsc_EpsUpdate_TaUpdate, // Table 22.5.18.3.3-17

                                                                              f_GutiParameters2MobileIdentity(tsc_IEI_Guti, v_Guti_Params3),

                                                                              cds_TAIListNonConsecutive_tlv (tsc_PLMN1, {tsc_TAC2}),

                                                                              v_EPS_ContextStatus,

                                                                              omit,//Lai

                                                                              omit,//MSId

                                                                              f_GetNBIOTAdditionalUpdateResult(v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType, CONTROL_PLANE),

                                                                              -, -, -,

                                                                              f_GetNBIOTNetworkFeatureSupport(CONTROL_PLANE),  // @sic R5-170563 sic@

                                                                              -, -, -,

                                                                              v_T3324Value2Mins)))); //T3324

    //@siclog "Step 95" siclog@

    //The UE transmits a TRACKING AREA UPDATE COMPLETE message

    SRB.receive(car_NB_SRB_NasPdu_IND(nbiot_Cell11,

                                      tsc_SRB1bis,

                                      cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                        cr_508_TAU_Complete)));

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 95");

    //@siclog "Step 96" siclog@

    //The SS releases the RRC connection.

    f_NBIOT_RRC_ConnectionRelease(nbiot_Cell11);

    t_T3324.start;
    t_T3324.timeout;

    //@siclog "Step 97" siclog@

    // When the T3324 timer expires the SS sends Paging message including a matched identity in a valid paging occasion as per normal DRX

    //@siclog "Step 98" siclog@

    // Check: Does the UE respond to the paging message?

    f_NBIOT_UE_Page_NoReaction (nbiot_Cell11,

                                cs_NB_Paging_OneRecord(cs_PagingUE_Identity_S_TMSI(v_Guti_Params3.MME_Code, v_Guti_Params3.M_TMSI)),

                                v_NoReactionTime);

    //@siclog "Step 99" siclog@

    // The UE is switched off

    f_UT_SwitchOffUE (UT, false);

    //@siclog "Step 100" siclog@

    // Set the cell type of Ncell 1 to the "Serving cell"

    // Set the cell type of Ncell 11 to the "Non-Suitable cell".

    v_CellPowerList := {

      cs_NB_CellPower(nbiot_Cell1, tsc_ServingCellRS_EPRE),

      cs_NB_CellPower(nbiot_Cell11, tsc_NonSuitableNBIOTCellRS_EPRE)

    };

    f_NBIOT_SetCellPowerList(v_CellPowerList);

    //@siclog "Step 101" siclog@

    f_UT_SwitchOnUE(UT, true); //Step1 - Switch on the UE

    //@siclog "Steps 102-114b1" siclog@

    // Steps 2 - 14b1 of the generic procedure precified in TS 36.508 subclause 8.1.5.2.3 is performed FFS: Step 14b1 handled on next step of prose

    fl_NBIOT_InitialRegistration2to13_eDRX_PSM(nbiot_Cell1, v_T3324Value2Mins, cs_EDRXParams('0000'B,'0011'B), -, v_Guti_Params3, v_Tai3); //Table 22.5.18.3.3-14
    //@siclog "Step 115" siclog@

    // Check: Does the UE transmit an ATTACH COMPLETE message including an Extended DRX parameters IE but without the T3324 IE?

    f_NBIOT_NAS_AttachComplete_CP (nbiot_Cell1, v_EpsDefaultBearerId, v_PDN.assignedPdnType);

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 115");

    f_NBIOT_RRC_ConnectionRelease(nbiot_Cell1);  

    //@siclog "Step 116" siclog@

    // SS sends a Paging message to the UE in a valid paging occasion in paging hyperframe as per Idle eDRX

    f_NBIOT_UE_Page_eDRX (nbiot_Cell1,

                          cs_NB_Paging_OneRecord(cs_PagingUE_Identity_S_TMSI(v_Guti_Params1.MME_Code, v_Guti_Params1.M_TMSI)),

                          0,  // PTW index value (2.56 sec)

                          3); // eDRX index value (40.96 sec)

    //@siclog "Step 117 - 120B" siclog@

    // Steps 117-118: Steps 2 to 3 of the NB-IoT Control Plane CIoT MT user data transfer non-SMS transport procedure (TS 36.508 8.1.5A.2.3) are executed.

    // Steps 119-120: Void

    // Step 120A : Check: Does the UE transmit a CONTROL PLANE SERVICE REQUEST according to step 4a1a1 or 4a1b1 in the generic procedure in TS 36.508 8.1.5A.2.3?

    // Step 120B: Steps 4a1b1 to 5 or only step 5 of the NB-IoT Control Plane CIoT MT user data transfer non-SMS transport procedure (TS 36.508 8.1.5A.2.3) are executed

    f_NBIOT_508Check_CP_ResponseToPagingForMTAccess_Step2_5(nbiot_Cell1);

    //@siclog "Step 121" siclog@

    //The SS releases the RRC connection.

    f_NBIOT_RRC_ConnectionRelease(nbiot_Cell1);

  } // function fl_tc_22_5_18_Body
2.2 Change 2 – Correction to function ‘fl_NBIOT_InitialRegistration2to13_eDRX_PSM’

	Function name
	fl_NBIOT_InitialRegistration2to13_eDRX_PSM

	Reason for change
	cdr_CIOT_ATTACH_REQUEST_WithExtdDRXParams_T3324 used in this function uses IMSI as the default Mobile ID parameter. The UE which has valid GUTI and TAI will fail at this function.

	Summary of change
	Open the Mobile ID parameter and TAI parameter in function fl_NBIOT_InitialRegistration2to13_eDRX_PSM

	TTCN module
	NBIOT_NAS_Attach.ttcn

	MCC160 Comment
	


Before change

  function fl_NBIOT_InitialRegistration2to13_eDRX_PSM(NBIOT_CellId_Type p_CellId,

                                                      template (omit) GPRS_Timer2 p_T3324 := omit,

                                                      template (omit) ExtdDRXParams p_EDRX := omit, //paramaters used for the attach accept message only

                                                      boolean p_NoPSM_and_eDRX := false  // indicates whether only eDRX params are taken into account on the attach accept message

                                                      ) runs on NBIOT_PTC

  {

    var template (present) NAS_UL_Message_Type v_ESMMessage;

    var NAS_UL_Message_Type v_ESMMessageRxd;

    var template (present) NAS_UL_Message_Type v_AttachReq;

    var boolean v_IPv4AllocationViaNasFlag := false;

    var template (omit) ExtdProtocolConfigOptions v_PcoToUE := omit;

    var template (omit) ExtdProtocolConfigOptions v_PcoFromUE := omit;

    var template (present) B3_Type v_PDNTypeRxd := (f_GetPdnType(), '101'B);

    var B3_Type v_PDNType;

    var NAS_MSG_Indication_Type v_NasInd;

    var NBIOT_PDN_Type v_PDN;

    //Steps 1-4 of the generic procedure specified in TS 36.508 subclause 8.1.5.2.3 are performed

    if (px_DoAttachWithoutPDN) {

      v_ESMMessage := cr_ESM_DUMMY_MESSAGE;

    } else {

      v_ESMMessage := cr_CIOT_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach, v_PDNTypeRxd);

    }

    v_AttachReq := cdr_CIOT_ATTACH_REQUEST_WithExtdDRXParams_T3324(cr_AdditionalUpdateType_CIOT,-, p_T3324); //All Att. Request messages on this test have "Any Value" as specific message contents

    // Receive RRC Connection Setup Complete with NAS

    v_NasInd := f_NBIOT_RRC_ConnEst_DefWithNas(p_CellId,

                                                cr_EstablishmentCause_NB_Any,

                                                cr_NAS_IndicationWithPiggybacking(tsc_SHT_IntegrityProtected,

                                                                                  v_AttachReq,

                                                                                  v_ESMMessage));

    v_ESMMessageRxd := v_NasInd.Pdu.PiggybackedPduList[0].Msg;

    //Do Steps 5-12

    f_NBIOT_InitialRegistration_Step5_12(p_CellId, CONTROL_PLANE, STATE3_NB_IDLEUPDATED, v_NasInd);

    if (not px_DoAttachWithoutPDN) {

      v_PDN := f_NBIOT_MobileInfo_GetAssignedPdn(0);  // This is registration, so this is the first PDN

      v_PcoFromUE := v_PDN.protocolConfigurationOptions;

      v_PDNType := v_PDN.assignedPdnType;

      if (match (v_PDNType, '101'B)) {

      } else if (isvalue(v_PcoFromUE)){

        v_IPv4AllocationViaNasFlag := f_CheckExtdPCOforIPallocationViaNas(v_PcoFromUE);

      }

      v_PcoToUE := f_GetDefaultExtdProtocolConfigOptions(v_PcoFromUE);

    }//ready for Step13

    // CP

    if (p_NoPSM_and_eDRX == true)

    {

       p_T3324 := omit; // case of TP13, the UE receives the ATTACH ACCEPT message including the extended DRX parameters but not the T3324 IE

    }

    //Step 13 of UE Registration acc. to 36.508 cl. 8.1.5.2

    f_NBIOT_InitialRegistration_Step13(nbiot_Cell1,

                                       CONTROL_PLANE,

                                       v_NasInd,

                                       v_IPv4AllocationViaNasFlag,

                                       v_PcoToUE,

                                       -,

                                       omit,

                                       omit,

                                       omit,

                                       omit,

                                       cs_GprsTimer_v_deact,

                                       omit,

                                       omit,

                                       p_T3324,

                                       omit,

                                       omit,

                                       p_EDRX);

    //Step 14

    //f_NBIOT_InitialRegistration_Step14(nbiot_Cell1, CONTROL_PLANE, tsc_EpsDefaultBearerId, v_PDNType);

    // not done as following step is done to Rx. Att. Accept message

  }
After change

   function fl_NBIOT_InitialRegistration2to13_eDRX_PSM(NBIOT_CellId_Type p_CellId,

                                                       template (omit) GPRS_Timer2 p_T3324 := omit,

                                                       template (omit) ExtdDRXParams p_EDRX := omit, //paramaters used for the attach accept message only

                                                       boolean p_NoPSM_and_eDRX := false,  // indicates whether only eDRX params are taken into account on the attach accept message

                                                       GutiParameters_Type p_Guti_Params,






       







template TrackingAreaId p_Tai) runs on NBIOT_PTC
  {

    var template (present) NAS_UL_Message_Type v_ESMMessage;

    var NAS_UL_Message_Type v_ESMMessageRxd;

    var template (present) NAS_UL_Message_Type v_AttachReq;

    var boolean v_IPv4AllocationViaNasFlag := false;

    var template (omit) ExtdProtocolConfigOptions v_PcoToUE := omit;

    var template (omit) ExtdProtocolConfigOptions v_PcoFromUE := omit;

    var template (present) B3_Type v_PDNTypeRxd := (f_GetPdnType(), '101'B);

    var B3_Type v_PDNType;

    var NAS_MSG_Indication_Type v_NasInd;

    var NBIOT_PDN_Type v_PDN;

    //Steps 1-4 of the generic procedure specified in TS 36.508 subclause 8.1.5.2.3 are performed

    if (px_DoAttachWithoutPDN) {

      v_ESMMessage := cr_ESM_DUMMY_MESSAGE;

    } else {

      v_ESMMessage := cr_CIOT_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach, v_PDNTypeRxd);

    }

    v_AttachReq := cdr_CIOT_ATTACH_REQUEST_WithExtdDRXParams_T3324(cr_AdditionalUpdateType_CIOT,f_GutiParameters2MobileIdentity(omit, p_Guti_Params), p_T3324, p_Tai); //All Att. Request messages on this test have "Any Value" as specific message contents

    // Receive RRC Connection Setup Complete with NAS

    v_NasInd := f_NBIOT_RRC_ConnEst_DefWithNas(p_CellId,

                                                cr_EstablishmentCause_NB_Any,

                                                cr_NAS_IndicationWithPiggybacking(tsc_SHT_IntegrityProtected,

                                                                                  v_AttachReq,

                                                                                  v_ESMMessage));

    v_ESMMessageRxd := v_NasInd.Pdu.PiggybackedPduList[0].Msg;

    //Do Steps 5-12

    f_NBIOT_InitialRegistration_Step5_12(p_CellId, CONTROL_PLANE, STATE3_NB_IDLEUPDATED, v_NasInd);

    if (not px_DoAttachWithoutPDN) {

      v_PDN := f_NBIOT_MobileInfo_GetAssignedPdn(0);  // This is registration, so this is the first PDN

      v_PcoFromUE := v_PDN.protocolConfigurationOptions;

      v_PDNType := v_PDN.assignedPdnType;

      if (match (v_PDNType, '101'B)) {

      } else if (isvalue(v_PcoFromUE)){

        v_IPv4AllocationViaNasFlag := f_CheckExtdPCOforIPallocationViaNas(v_PcoFromUE);

      }

      v_PcoToUE := f_GetDefaultExtdProtocolConfigOptions(v_PcoFromUE);

    }//ready for Step13

    // CP

    if (p_NoPSM_and_eDRX == true)

    {

       p_T3324 := omit; // case of TP13, the UE receives the ATTACH ACCEPT message including the extended DRX parameters but not the T3324 IE

    }

    //Step 13 of UE Registration acc. to 36.508 cl. 8.1.5.2

    f_NBIOT_InitialRegistration_Step13(nbiot_Cell1,

                                       CONTROL_PLANE,

                                       v_NasInd,

                                       v_IPv4AllocationViaNasFlag,

                                       v_PcoToUE,

                                       -,

                                       omit,

                                       omit,

                                       omit,

                                       omit,

                                       cs_GprsTimer_v_deact,

                                       omit,

                                       omit,

                                       p_T3324,

                                       omit,

                                       omit,

                                       p_EDRX);

    //Step 14

    //f_NBIOT_InitialRegistration_Step14(nbiot_Cell1, CONTROL_PLANE, tsc_EpsDefaultBearerId, v_PDNType);

    // not done as following step is done to Rx. Att. Accept message

  }
2.3 Change 3 – Correction to template ‘cdr_CIOT_ATTACH_REQUEST_WithExtdDRXParams_T3324’

	Template name
	cdr_CIOT_ATTACH_REQUEST_WithExtdDRXParams_T3324

	Reason for change
	Incorrect NAS key set identifier.

	Summary of change
	Since the status of context in UE is uncertain, NAS key set ID should be an allowed value rather than ‘no key’(‘111’).

	TTCN module
	NBIOT_NAS_Attach.ttcn

	MCC160 Comment
	


Before change

  template (present) NAS_UL_Message_Type cdr_CIOT_ATTACH_REQUEST_WithExtdDRXParams_T3324(template AdditionalUpdateType p_AdditionalUpdateType,

                                                                                         template (present) MobileIdentity p_MobileId := f_Imsi2MobileIdentity(px_IMSI_Def),

                                                                                         template GPRS_Timer2 p_T3324 := omit,

                                                                                         template TrackingAreaId p_Tai := omit,

                                                                                         template ExtdDRXParams p_EDRX := cr_ExtdDRXParamsAny)

    modifies cr_CIOT_ATTACH_REQUEST :=

  {

    aTTACH_REQUEST := {

      nasKeySetId              := cr_NAS_KeySetIdentifier(tsc_NasKsi_NoKey, tsc_Spare1),

      epsMobileIdentity        := p_MobileId,

      lastVisitedRegisteredTai := p_Tai,

      t3324Value               := p_T3324,

      extendedDRXParams        := p_EDRX

    }

  };
After change

    template (present) NAS_UL_Message_Type cdr_CIOT_ATTACH_REQUEST_WithExtdDRXParams_T3324(template AdditionalUpdateType p_AdditionalUpdateType,

                                                                                         template (present) MobileIdentity p_MobileId := f_Imsi2MobileIdentity(px_IMSI_Def),

                                                                                         template GPRS_Timer2 p_T3324 := omit,

                                                                                         template TrackingAreaId p_Tai := omit,

                                                                                         template ExtdDRXParams p_EDRX := cr_ExtdDRXParamsAny)

    modifies cr_CIOT_ATTACH_REQUEST :=

  {

    aTTACH_REQUEST := {

      nasKeySetId              := cr_NAS_KeySetIdentifier(?, tsc_Spare1),

      epsMobileIdentity        := p_MobileId,

      lastVisitedRegisteredTai := p_Tai,

      t3324Value               := p_T3324,

      extendedDRXParams        := p_EDRX

    }

  };
2.4 Change 4 – Correction to function ‘f_TC_22_5_18_NBIOT’
	Function name
	f_TC_22_5_18_NBIOT

	Reason for adding
	Since Ncell2 has the same TA with Ncell1, the UE will not transimit TAU request at step 24. The related prose CR will be present in the next RAN#76 meeting. .

	Summary of change
	Use Ncell11 instead of Ncell 2 in this test case.

	TTCN module
	NBIOT_NAS_Attach.ttcn

	MCC160 Comment
	


Before Change
  function f_TC_22_5_18_NBIOT() runs on NBIOT_PTC

  { //NB-IoT / Attach & Normal tracking area update Procedure / Success / without Idle eDRX parameters /

    // With Idle eDRX parameters/ With and without Idle eDRX and PSM parameters

    // @sic R5-171207 sic@

    var template (value) GPRS_Timer2 v_T3324Value2Mins := cs_GPRS_Timer2_3_IEI('6A'O, tsc_GprsTimerUnit_1min, '00010'B);  //value = 2 mins

    f_NBIOT_Init(c1);

    // create cell and send out system information

    f_NBIOT_CellConfig_Def(nbiot_Cell1, CONTROL_PLANE);

    f_NBIOT_CellConfig_Def(nbiot_Cell2, CONTROL_PLANE);

    // The UE is configured to use Power Saving Mode

    f_UT_MMI_ConfigurePSM(UT, true);

    f_UT_MMI_ConfigureEDRX(UT, true);

    // Preamble: The UE is in State 1-NB "Switched OFF"

    // The UE is previously registered on E-UTRAN, and when on E-UTRAN, the UE is last authenticated and registered on the NB-IoT cell using default message contents according to TS 36.508 [18]

    // UE is configured to request the use of eDRX (in the ATTACH REQUEST and TRACKING AREA UPDATE messages)

    // The UE is configured to set T3324 to two minutes

    // ==========================================================================

    f_NBIOT_Preamble(nbiot_Cell1, CONTROL_PLANE);

    f_UT_ConfigurePSM_Value(UT, omit, omit, v_T3324Value2Mins);

    f_NBIOT_UE_Detach_SwitchOffUe(nbiot_Cell1, CONTROL_PLANE, RRC_CONNECTED);

    f_NBIOT_SetCellPower(nbiot_Cell2, tsc_NonSuitableNBIOTCellRS_EPRE);

    f_NBIOT_TestBody_Set(true);

    fl_TC_22_5_18_Body();

    // Postamble

    f_NBIOT_TestBody_Set(false);

    f_NBIOT_Postamble (nbiot_Cell1, CONTROL_PLANE, N1_IDLE);

  }
After change

  function f_TC_22_5_18_NBIOT() runs on NBIOT_PTC

  { //NB-IoT / Attach & Normal tracking area update Procedure / Success / without Idle eDRX parameters /

    // With Idle eDRX parameters/ With and without Idle eDRX and PSM parameters

    // @sic R5-171207 sic@

    var template (value) GPRS_Timer2 v_T3324Value2Mins := cs_GPRS_Timer2_3_IEI('6A'O, tsc_GprsTimerUnit_1min, '00010'B);  //value = 2 mins

    f_NBIOT_Init(c1);

    // create cell and send out system information

    f_NBIOT_CellConfig_Def(nbiot_Cell1, CONTROL_PLANE);

    f_NBIOT_CellConfig_Def(nbiot_Cell11, CONTROL_PLANE);

    // The UE is configured to use Power Saving Mode

    f_UT_MMI_ConfigurePSM(UT, true);

    f_UT_MMI_ConfigureEDRX(UT, true);

    // Preamble: The UE is in State 1-NB "Switched OFF"

    // The UE is previously registered on E-UTRAN, and when on E-UTRAN, the UE is last authenticated and registered on the NB-IoT cell using default message contents according to TS 36.508 [18]

    // UE is configured to request the use of eDRX (in the ATTACH REQUEST and TRACKING AREA UPDATE messages)

    // The UE is configured to set T3324 to two minutes

    // ==========================================================================

    f_NBIOT_Preamble(nbiot_Cell1, CONTROL_PLANE);

    f_UT_ConfigurePSM_Value(UT, omit, omit, v_T3324Value2Mins);

    f_NBIOT_UE_Detach_SwitchOffUe(nbiot_Cell1, CONTROL_PLANE, RRC_CONNECTED);

    f_NBIOT_SetCellPower(nbiot_Cell11, tsc_NonSuitableNBIOTCellRS_EPRE);

    f_NBIOT_TestBody_Set(true);

    fl_TC_22_5_18_Body();

    // Postamble

    f_NBIOT_TestBody_Set(false);

    f_NBIOT_Postamble (nbiot_Cell1, CONTROL_PLANE, N1_IDLE);

  }

}
2. Branches executed

None.

3. Execution Log Files

4.1 Hisilicon Hi2110

The Hisilicon Hi2110 passed this test case on Starpoint SP8630 Integration Test System in LTE FDD band 8. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file
TC_22_5_18_Hi2110_PASS.spm
In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

· PICS/PIXIT file name
TC_22_5_18.sppar

4. References

	[1]
	R5-170497: Supporting information for agreement of NB-IoT test case 22.5.18. This archive comprises html and text format execution log files and PICS/PIXIT settings file respectively.


